

Master Maths 1 2009-2010
Optimisation
correction TP1

Exercice 0 *Introduction*

Ecrire un script scilab (que vous pourrez appeler `exo1.sce`), qui :

1. permet de définir une valeur à l'angle θ ,
2. permet de définir la matrice A et le vecteur x ,
3. calcule $y = Ax$.

Réponse :

```
theta = %pi/4 // par exemple
A = [ cos(theta), -sin(theta) ; ...
      sin(theta),  cos(theta) ]
x = [1;1] // par exemple
y = A*x
```

Expérimenter votre script avec plusieurs valeurs pour θ et pour le vecteur x . Comment :

1. extraire la deuxième composante de x ? *Réponse :* `x(2)`
2. extraire l'élément $a_{1,2}$? *Réponse :* `A(1,2)`
3. extraire la première ligne de A ? *Réponse :* `A(1,:)`
4. extraire la deuxième colonne de A ? *Réponse :* `A(:,2)`

Exercice 1 *Concaténation de matrices*

Continuer le script précédent en :

1. définissant un vecteur v de format $(1,2)$ rempli de zéros ;
2. puis en définissant par concaténation la matrice :

$$R = \begin{bmatrix} \cos(\theta) & -\sin(\theta) & 0 \\ \sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix},$$

à l'aide de la matrice A et du vecteur v précédemment définis.

Réponse :

```
v = zeros(1,2)
R = [ A, v'; ...
      v, 1 ]
```

Exercice 2 *Encore une matrice spéciale*

Définir (dans un script ou directement) la matrice d'ordre n suivante :

$$A = \begin{pmatrix} 2 & -1 & 0 & \cdots & 0 \\ -1 & \ddots & \ddots & \ddots & \vdots \\ 0 & \ddots & \ddots & \ddots & 0 \\ \vdots & \ddots & \ddots & \ddots & -1 \\ 0 & \cdots & 0 & -1 & 2 \end{pmatrix}$$

Réponse :

```

n = 6 // par exemple
v = ones(1,n-1)
A = 2*eye(n,n) - diag(v,1) - diag(v,-1)

```

Exercice 3 *extractions*

- Définir avec `rand` un vecteur ligne à 5 composantes (c-a-d une matrice de format (1,5) et le renverser (la dernière composante devient la première, l'avant dernière devient la deuxième, ...) via une extraction. *Réponse :*

```

x = rand(1,5)
x = x(5:-1:1) // on peut aussi utiliser x($:-1:1)

```

- Définir une matrice de format (6,5) avec `rand` puis en extraire la sous-matrice formée par l'intersection des lignes 1,2 et 6 et des colonnes 2,4,5. *Réponse :*

```

A = rand(6,5)
A([1,2,6],[2,4,5])

```

Exercice 4 *linspace, opérations élément par élément*

- Définir un vecteur x avec 100 composantes permettant de mailler régulièrement l'intervalle $[0, 2\pi]$ (une variable prédéfinie `%pi` contient une approximation de π); définir ensuite le vecteur y tel que $y_i = \sin(x_i)$ puis tracer la courbe (cf mini-tutoriel graphique à la suite de cet exercice). *Réponse :*

```

x = linspace(0,2*%pi,100);
y = sin(x);
scf(0) // la figure graphique courante est la figure 0
clf()
plot(x,y,"b")
legend("sin(x)")

```

- Même question avec successivement :

$$f_2(x) = (x-1)(x+1), x \in [-2, 2]$$

$$f_3(x) = \frac{1}{1+x^2}, x \in [-3, 3]$$

$$f_4(x) = \cos\left(\frac{1}{0.1+x^2}\right), x \in [-3, 3]$$

Vous pouvez adapter le nombre de points du maillage au besoin. Représenter f_3 et f_4 sur la même figure. *Réponse :*

```

x2 = linspace(-2,2,100);
y2 = (x2-1).*(x2+1);
scf(1) // la figure graphique courante est la figure 1
clf()
plot(x2,y2,"r")
legend("(x-1)(x+1)")
x3 = linspace(-3,3,1000);
y3 = 1 ./ (1 + x3.^2); // bien mettre un espace entre 1 et ./
y4 = cos(1 ./ (0.1 + x3.^2)); // même remarque
scf(2)
clf()
plot(x3,y3,"b",x3,y4,"r")
legend("1/(1+x^2)", "cos(1/(0.1 + x^2))")
xtitle("le titre...")

```

Exercice 5 *Ecrire des fonctions Scilab*

1. Ecrire une fonction pour résoudre une équation du second degré. *Réponse :*

```
function [x1,x2] = resoud_equ_2d(a, b, c)
    // calcul des racines de a x^2 + b x + c = 0
    // a, b et c peuvent etre des reels ou des complexes et a doit etre non nul
    if a == 0 then
        error("a ne doit pas être nul !")
    end
    delta = b^2 - 4*a*c;
    x1 = (-b - sqrt(delta))/(2*a)
    x2 = (-b + sqrt(delta))/(2*a)
endfunction
```

2. Ecrire une fonction `somme` qui fait la somme des composantes d'un vecteur à l'aide d'une boucle (c-a-d ne pas utiliser la fonction `sum` de Scilab; par contre cette fonction devra être utilisée dans tous les autres cas!). *Réponse :*

```
function [s] = somme(x)
    s = 0;
    for i = 1:length(x)
        s = s + x(i);
    end
endfunction
```

3. Ecrire une fonction qui renvoie la moyenne et l'écart type d'un vecteur. Tester avec la fonction `rand`. On rappelle que $E[X] = 0.5$ et $\sigma[X] = \sqrt{1/12}$ si $X \sim U(0,1)$. *Réponse :*

```
function [mu,sigma] = moyenne_et_ecart_type(x)
    n = length(x);
    mu = sum(x)/n;
    sigma = sqrt( sum( (x - mu).^2 )/(n-1) )
endfunction
```